
DUEMMEGI 

SISTEMI DI AUTOMAZIONE INDUSTRIALE

CONTATTO

**Raccolta di Note Applicative ed esempi di
programmazione per MCP**

DUEMMEGI s.r.l. - Via Longhena, 4 - 20139 MILANO
Tel. 02/57300377 - FAX 02/55213686

Indice

Gestione salita e discesa tapparella motorizzata mediante MCP	3
Gestione di una sirena di allarme con tacitazione mediante MCP	4
Gestione rotazione di una ruota mediante prossimetro: controllo del teleruttore di comando rotazione e generazione dell' allarme di ruota ferma con MCP	5
Gestione di 3 pompe gemellari a rotazione temporale mediante MCP	6
Gestione di 2 pompe gemellari a rotazione temporale mediante MCP	8
Controllo del livello di un serbatotio con due pompe a funzionamento alternato	10
Gestione 2 pompe gemellari con pressostato mediante MCP (Versione=<12).....	11
Conversione da codice binario a uscite dirette	12
Pulsanti passo-passo locali e un pulsante generale di ON-OFF	15
Pulsanti passo-passo locali, pulsanti di gruppo e un pulsante generale di ON-OFF	16
Guida alla configurazione dei modem in un sistema PC – MCP collegati via linea telefonica	18
Gestione di sequenze allarmi con MCP e DISPBUS	23
Termoregolazione ambientale mediante modulo CLIMA e MCP	25
Gestione di una uscita analogica per controllare un dimmer.....	27
Esempi di messaggi per la realizzazione di un driver di comunicazione con MCP (protocollo FXP).....	28

Gestione salita e discesa tapparella motorizzata mediante MCP

```

////////////////////////////////////
////// GESTIONE SALITA E DISCESA TAPPARELLA MOTORIZZATA MEDIANTE MCP ////
////////////////////////////////////
// NELL' ESEMPIO CHE SEGUE SONO STATI ASSUNTI:
// I1.1 = INGRESSO PULSANTE SALITA TAPPARELLA
// I1.2 = INGRESSO PULSANTE DISCESA TAPPARELLA
// O1.1 = USCITA COMANDO RELE' PER SALITA TAPPARELLA
// O1.2 = USCITA COMANDO RELE' PER DISCESA TAPPARELLA
////////////////////////////////////
// DESCRIZIONE DI FUNZIONAMENTO:
// PER FAR PARTIRE LA SALITA TAPPARELLA, MANTENERE PREMUTO IL PULSANTE
// DI SALITA PER ALMENO 0.5 SECONDI; DOPO QUESTO TEMPO LA TAPPARELLA
// CONTINUA LA CORSA ANCHE AL RILASCIO DEL PULSANTE. PER ARRESTARE IL
// MOVIMENTO PRIMA DEL RAGGIUNGIMENTO DEL FONDO CORSA, PREMERE IL
// PULSANTE DI DISCESA PER UN TEMPO INFERIORE A 0.5 SECONDI.
// SE IL PULSANTE DI DISCESA VIENE INVECE TENUTO PREMUTO PER UN TEMPO
// SUPERIORE A 0.5 SECONDI, SI OTTIENE LA PARTENZA DEL MOVIMENTO DI
// DISCESA. IN ALTRE PAROLE, UNA BREVE PRESSIONE DEL PULSANTE EQUIVALE
// A UN CONTROCOMANDO (STOP), MENTRE UNA PRESSIONE PIU' PROLUNGATA
// EQUIVALE AD UN COMANDO NELLA DIREZIONE DEFINITA DA QUEL PULSANTE.
// PER IL MOVIMENTO DI DISCESA VALE QUANTO DETTO SOPRA, SCAMBIANDO IL
// PULSANTE DI DISCESA CON QUELLO DI SALITA.
// IL SISTEMA QUI DESCRITTO PREVEDE UN TEMPO MASSIMO DOPO IL QUALE
// I RELE' VENGONO COMUNQUE DISECCITATI; IN QUESTO MODO VIENE PROTETTO
// IL MOTORE NEL CASO IN CUI IL FINECORSO DELLA TAPPARELLA SI DOVESSE
// GUASTARE OPPURE SE SI INCONTRA UN OSTACOLO DURANTE LA CORSA.
// QUESTO TEMPO MASSIMO (TIMEOUT) NELL' ESEMPIO CHE SEGUE E'
// DEFINITO DALLA V2 E DALLA V4 E VALE 5 SECONDI (PER CAMBIARLO E'
// SUFFICIENTE CAMBIARE IL 50 NELLA DEFINIZIONE DI V2 E DI V4, TENENDO
// PRESENTE CHE IL NUMERO SCRITTO E' IN DECIMI DI SECONDO, CIOE' 50
// SIGNIFICA 5 SECONDI). IL VALORE DA ASSEGNARE A QUESTO TEMPO DIPENDE
// DAL SISTEMA DI MOVIMENTAZIONE UTILIZZATO; SI CONSIGLIA DI ASSEGNARE
// A QUESTO TEMPO UN VALORE PARI A CIRCA 1,5 VOLTE IL TEMPO IMPIEGATO
// DALLA TAPPARELLA AD ESEGUIRE UNA CORSA COMPLETA.
// V1 E V3 DEFINISCONO INVECE IL TEMPO MINIMO DI PRESSIONE DEI PULSANTI
// PERCHE' IL COMANDO VENGA ESEGUITO COME DESCRITTO SOPRA. IL VALORE 5
// SIGNIFICA 5 DECIMI DI SECONDO (0.5 SECONDI) E DOVREBBE GIA' ESSERE UN
// VALORE OTTIMALE PER QUESTO TIPO DI APPLICAZIONE.
////////////////////////////////////
// EQUAZIONI NECESSARIE PER LA GESTIONE DI UNA TAPPARELLA:

V1 = TIMER(I1.1,5,0)
V2 = TIMER(I1.1,0,50)
V3 = TIMER(I1.2,5,0)
V4 = TIMER(I1.2,0,50)
O1.1 = SV1 & SI1.1 & RI1.2 & V2
O1.2 = SV3 & SI1.2 & RI1.1 & V4

```

Gestione di una sirena di allarme con tacitazione mediante MCP

```

////////////////////////////////////
////// GESTIONE DI UN RELE' DI UNA SIRENA DI ALLARME CON TACITAZIONE ////
////// MEDIANTE MCP //////////////////////////////////////
////////////////////////////////////
// DESCRIZIONE DI FUNZIONAMENTO:
////////////////////////////////////
// DATI ALCUNI CONTATTI DI ALLARME, SI VUOLE CHE ALLA CHIUSURA DI UNO
// QUALSIASI DI ESSI SI ATTIVI UNA USCITA CHE CONTROLLA UNA SIRENA. LA
// SIRENA DEVE RIMANERE ALIMENTATA ANCHE NEL CASO IN CUI IL CONTATTO CHE
// HA PROVOCATO L'ALLARME SI RIAPRE (FUNZIONE DI MEMORIZZAZIONE).
// ALLA PRESSIONE DI UN PULSANTE, SI VUOLE CHE LA SIRENA VENGA TACITATA,
// INDIPENDENTEMENTE DAL FATTO CHE IL CONTATTO CHE HA PROVOCATO L'ALLARME
// SIA TORNATO O MENO ALLO STATO DI RIPOSO. SE UN ALTRO CONTATTO DI
// ALLARME, OPPURE LO STESSO DI CUI SOPRA, SI CHIUDE DOPO AVER TACITATO LA
// SIRENA, QUESTA VIENE RIATTIVATA. NELL' ESEMPIO CHE SEGUE SONO STATI ASSUNTI:
// I1.1 = INGRESSO ALLARME 1
// I1.2 = INGRESSO ALLARME 2
// I1.3 = INGRESSO ALLARME 3
// I1.4 = INGRESSO ALLARME 4
// I1.8 = INGRESSO PULSANTE DI RESET
// O1.1 = USCITA COMANDO SIRENA
// V1 = PUNTO VIRTUALE DI APOGGIO
// SI E' ASSUNTO INOLTRE, PER ESEMPIO, CHE L'ALLARME DEBBA ESSERE GENERATO
// ALLA CHIUSURA DI I1.1, I1.3 E I1.4 O ALL' APERTURA DI I1.2 (IN ALTRE PAROLE
// I CONTATTI DI ALLARME SONO TUTTI NORMALMENTE APERTI AD ECCEZIONE DI I1.2
// CHE E' NORMALMENTE CHIUSO). // LE EQUAZIONI NECESSARIE PER REALIZZARE QUANTO
// DESCRITTO SONO:

V1 = TI1.1 | T!I1.2 | TI1.3 | TI1.4 | RI1.8
O1.1 = SV1 & RI1.8

// GLI INGRESSI NELLA EQUAZIONE SOPRA RIPORTATA POSSONO ANCHE ESSERE VIRTUALI.

// IN ALTERNATIVA E' POSSIBILE IMPLEMENTARE LA FUNZIONE QUI DESCRITTA
// UTILIZZANDO IL BLOCCO 'SERIAL' IN MODO UN PO' "ANOMALO"; INFATTI PER IL
// FUNZIONAMENTO DEL BLOCCO 'SERIAL' NON E' NECESSARIO CHE UN DISPLAY SERIALE
// SIA EFFETTIVAMENTE COLLEGATO ALLA APPOSITA PORTA DI MCP.
// NEL SEGUENTE ESEMPIO SI UTILIZZA QUESTA TECNICA PER IMPLEMENTARE
// L'APPLICAZIONE RICHIESTA. LE EQUAZIONI NECESSARIE IN QUESTO CASO SONO:

V1 = I1.1 // INGR. ALL. 1
V2 = I1.2 // INGR. ALL. 2
V3 = I1.3 // INGR. ALL. 3
V4 = I1.4 // INGR. ALL. 4
V100 = I1.8 // I1.8 INGR. PULSANTE TAC. SIRENA
O1.1 = V101 // USCITA COMANDO SIRENA
SERIAL ( \
    ACK = V100 \
    SIREN = V101 \
    DEFAULT = NOMEM NOPRN SIREN \
    Z001 = V1 \ // NEL BLOCCO SERIAL VANNO INSERITI
    Z002 = V2 \ // GLI ALLARMI TACITABILI
    Z003 = V3 \ // DAL PULSANTE DEFINITO DALLA V100
    Z004 = V4 \
)

```

Gestione rotazione di una ruota mediante prossimetro: controllo del teleruttore di comando rotazione e generazione dell' allarme di ruota ferma con MCP

```

////////////////////////////////////
// GESTIONE ROTAZIONE DI UNA RUOTA MEDIANTE PROSSIMETRO: CONTROLLO DEL
// TELERUTTORE DI COMANDO ROTAZIONE E GENERAZIONE DELL' ALLARME DI RUOTA
// FERMA CON MCP //////////////////////////////////////
////////////////////////////////////
// DESCRIZIONE DI FUNZIONAMENTO:
////////////////////////////////////
// IL SISTEMA QUI DESCRITTO DEVE GENERARE LA PARTENZA DI UN MOTORE CHE
// METTE IN MOVIMENTO UNA RUOTA ALLA CHIUSURA DI ALCUNI CONTATTI E
// ALL' APERTURA DI ALTRI. NEL CASO IN CUI LA RUOTA SI BLOCCA (AD ESEMPIO
// PERCHE' SI SONO ROTTI GLI ORGANI DI TRASMISSIONE DEL MOTO) IL SISTEMA
// DEVE SPEGNERE IL TELERUTTORE CHE COMANDA IL MOTORE E DEVE SEGNALARE UN
// ALLARME SU UN DISPLAY. IL SISTEMA E' COMPLETATO DA UN PROSSIMETRO CHE
// GENERA UN IMPULSO AD OGNI GIRO COMPLETO DELLA RUOTA.
// NELL' ESEMPIO CHE SEGUE SONO STATI ASSUNTI:
// I1.1 = CONTATTO N.A. DI CONSENSO MARCIA (IL MOTORE PARTE ALLA CHIUSURA)
// I1.2 = CONTATTO N.C. DI CONSENSO MARCIA (IL MOTORE PARTE ALLA APERTURA)
// I1.3 = CONTATTO N.A. DI CONSENSO MARCIA (IL MOTORE PARTE ALLA CHIUSURA)
// I1.4 = CONTATTO N.A. DI CONSENSO MARCIA (IL MOTORE PARTE ALLA CHIUSURA)
// I1.7 = TESTIMONE N.A. DEL TELERUTTORE (CHIUDE ALLA ATTIV. DEL TELERUTT.)
// I1.8 = INGRESSO FOTOCELLULA
// O1.8 = USCITA COMANDO TELERUTTORE MOTORE
// IL TEMPO (30) RIPORTATO NELLE EQUAZIONI TIMER CHE SEGUONO DEVE ESSERE CIRCA
// 1.2 VOLTE IL TEMPO NECESSARIO AD EFFETTUARE UN GIRO COMPLETO DELLA RUOTA.
// LE EQUAZIONI NECESSARIE PER REALIZZARE QUANTO DESCRITTO SONO:

V1 = TIMER (!I1.1,0,30) // I1.1 = INGRESSO N.A. DI CONSENSO MARCIA
V2 = TIMER (I1.2,30,0) // I1.2 = INGRESSO N.C. DI CONSENSO MARCIA
V3 = TIMER (!I1.3,0,30) // I1.1 = INGRESSO N.A. DI CONSENSO MARCIA
V4 = TIMER (!I1.4,0,30) // I1.1 = INGRESSO N.A. DI CONSENSO MARCIA

V5 = I1.1 & !I1.2 & I1.3 & I1.4 // V5 DIVENTA ATTIVA QUANDO TUTTI I
// CONSENSI SONO ATTIVI

V6 = TIMER (I1.8,0,30) // I1.8 E' L'INGRESSO DELLA FOTOCELLULA
V7 = TIMER (!I1.8,0,30) // I1.8 E' L'INGRESSO DELLA FOTOCELLULA
// IL TEMPO INDICATO NELLE EQUAZIONI SOPRA
// DEVE ESSERE MAGGIORE DEL TEMPO NECESSARIO
// AD EFFETTUARE UN GIRO COMPLETO (CIRCA 1.2 VOLTE)

V127 = T!I1.7 | R!I1.1 | RI1.2 | R!I1.3 | R!I1.4 // V127 E' IL VIRTUALE
// CHE FA USCIRE SUL
// DISPLAY IL MESSAGGIO DI
// RUOTA FERMA.
// I1.7 E' IL TESTIMONE DEL
// TELERUTTORE

O1.8 = V5 & (V1 | V2 | V3 | V4) | (V5 & V6 & V7) // USCITA DI COMANDO
// TELERUTTORE

```

Gestione di 3 pompe gemellari a rotazione temporale mediante MCP

Funzionamento:

Ogni pompa e' controllata da un commutatore AUTO-0-MAN; il funzionamento del sistema dipende dalle posizioni di questi 3 commutatori come qui di seguito descritto.

In posizione AUTO: devono essere sempre accese almeno 2 pompe e deve avvenire lo scambio automatico ad ogni determinato periodo di tempo (es. 1 settimana).

La sequenza di accensione delle pompe, allo scadere di ogni settimana, e' la seguente:

P1	P2	P3
ON	ON	OFF
OFF	ON	ON
ON	OFF	ON

dopo di che la sequenza riprende dall'inizio.

Se si apre il contatto di anomalia (es. contatto termico) di una delle pompe in funzione, la stessa pompa si spegne e si accende quella di riserva (vale a dire quella che in quel momento era spenta); da questo momento lo scambio a tempo rimane disabilitato sino a che il contatto di anomalia non si richiude.

Se almeno due o piu' pompe sono spente contemporaneamente (per anomalia o per qualsiasi altro motivo), si deve generare un allarme attivando una uscita.

In posizione 0: se si porta il commutatore di controllo di una pompa che in quel momento e' accesa su 0, deve partire la pompa di riserva (a patto che le altre due pompe siano in AUTO). Se si porta un altro commutatore su 0 rimane attiva solo la pompa in AUTO e si attiva l'uscita di allarme che segnala che almeno 2 pompe sono ferme. Se si porta anche il terzo commutatore su 0 si spegne anche la terza pompa e rimane attivo l' allarme.

In posizione MAN: portando un commutatore in questa posizione, la pompa ad esso associata si accende (a meno che non sia in anomalia); se le altre due pompe sono in automatico, solo una delle due risultera' accesa (in quanto almeno due ma non piu' di due devono essere accese contemporaneamente).

Se tutti i commutatori sono in manuale, risulteranno accese tutte le pompe, almeno sino a quando non scatta uno o piu' contatti di anomalia.

Realizzazione mediante MCP:

Per l'implementazione di questo automatismo con MCP sono necessari:

- 3 punti di ingresso collegati ai testimoni dei teleruttori di ognuna delle 3 pompe (normalmente aperti)
- 3 punti di ingresso collegati ai contatti di anomalia di ognuna delle 3 pompe (normalmente chiusi)
- 2 punti di ingresso collegati alle posizioni AUTO e MAN del commutatore che controlla la pompa 1
- 2 punti di ingresso collegati alle posizioni AUTO e MAN del commutatore che controlla la pompa 2
- 2 punti di ingresso collegati alle posizioni AUTO e MAN del commutatore che controlla la pompa 3
- 3 punti di uscita per il comando dei teleruttori di ognuna delle 3 pompe
- 1 punto di uscita per l'attivazione dell'allarme

Le equazioni che realizzano l'automatismo sopra descritto sono qui di seguito riportate ipotizzando indirizzi e punti a titolo esemplificativo. Ad ognuno di questi indirizzi e punti sara' sufficiente sostituire quelli dell'impianto reale per ottenere le equazioni volute.

```

////////////////////////////////////
// SEQUENZA 3 POMPE GEMELLARI //////////////////////////////////////
////////////////////////////////////
// PUNTI DI INGRESSO E USCITA UTILIZZATI E LORO SIGNIFICATO:
// V5 = CLOCK SUL QUALE E' PRESENTE 1 IMPULSO ALLA SETTIMANA
// I1.1 = TESTIMONE P1
// I1.2 = TESTIMONE P2
// I1.3 = TESTIMONE P3
// I1.5 = ANOMALIA P1
// I1.6 = ANOMALIA P2
// I1.7 = ANOMALIA P3
// I2.1 = P1 AUTO
// I2.2 = P1 MAN
// I2.3 = P2 AUTO
// I2.4 = P2 MAN
// I2.5 = P3 AUTO
// I2.6 = P3 MAN
// O1.1 = P1
// O1.2 = P2
// O1.3 = P3
// O1.8 = ALLARME

V5 = CLOCK(LUN:00:00,LUN:00:01) // 1 IMPULSO DA 1 MINUTO
// ALLA SETTIMANA

V1 = C = 1,3 UV5 // VIRTUALE PER SCAMBIO
V2 = C = 2,3 UV5 // VIRTUALE PER SCAMBIO

O1.1 = I2.1 & I1.5 & ( \ // BLOCCO P1 IN AUTO:
                !V1 & !I2.6 | \ // ON DA CLOCK
                !I1.6 | !I1.7 | \ // ON DA ANOM. P2 o P3
                !I2.6 & (!I2.3 | !I2.5) \ // ON SE P2 o P3 SU 0
                ) | \
        I2.2 & I1.5 // P1 IN MANUALE

O1.2 = I2.3 & I1.6 & ( \ // BLOCCO P2 IN AUTO:
                !V2 & !I2.2 | \ // ON DA CLOCK
                !I1.5 | !I1.7 | \ // ON DA ANOM. P1 o P3
                !I2.2 & (!I2.1 | !I2.5) \ // ON SE P1 o P3 SU 0
                ) | \
        I2.4 & I1.6 // P2 IN MANUALE

O1.3 = I2.5 & I1.7 & ( \ // BLOCCO P3 IN AUTO:
                ((!V1 & V2) | (V1 & !V2)) & !I2.4 | \ // ON DA CLOCK
                !I1.5 | !I1.6 | \ // ON DA ANOM. P1 o P2
                !I2.4 & (!I2.1 | !I2.3) \ // ON SE P1 o P2 SU 0
                ) | \
        I2.6 & I1.7 // P3 IN MANUALE

O1.8 = (!I1.1 & !I1.2) | \ // USCITA ALLARME: ATTIVO
        (!I1.2 & !I1.3) | \ // QUANDO CI SONO ALMENO
        (!I1.3 & !I1.1) // 2 POMPE FERME

```

Gestione di 2 pompe gemellari a rotazione temporale mediante MCP

Funzionamento:

Ogni pompa e' controllata da un commutatore AUTO-0-MAN; il funzionamento del sistema dipende dalle posizioni di questi 2 commutatori come qui di seguito descritto.

In posizione AUTO: devono essere sempre accese almeno 1 pompa e deve avvenire lo scambio automatico ad ogni determinato periodo di tempo (es. 1 settimana).

La sequenza di accensione delle pompe, allo scadere di ogni settimana, e' la seguente:

P1	P2
ON	OFF
OFF	ON

dopo di che la sequenza riprende dall'inizio.

Se si apre il contatto di anomalia (es. contatto termico) della pompa in funzione, la stessa pompa si spegne e si accende quella di riserva (vale a dire quella che in quel momento era spenta); da questo momento lo scambio a tempo rimane disabilitato sino a che il contatto di anomalia non si richiude.

Se entrambe le pompe si spengono (per lo scatto di entrambe le anomalie, si deve generare un allarme attivando una uscita.

In posizione 0: se si porta il commutatore di controllo della pompa che in quel momento e' accesa su 0, deve partire la pompa di riserva (a patto che questa sia in AUTO). Se si porta anche l' altro commutatore su 0 si spegne anche la seconda e si attiva l' uscita di allarme.

In posizione MAN: portando un commutatore in questa posizione, la pompa ad esso associata si accende (a meno che non sia in anomalia); se l'altra pompa e' in automatico, essa verra' automaticamente spenta (in quanto almeno una ma non piu' di una pompa devono essere accese contemporaneamente).

Se entrambi i commutatori sono in manuale, risulteranno accese entrambe le pompe, almeno sino a quando non scatta uno o piu' contatti di anomalia.

Realizzazione mediante MCP:

Per l'implementazione di questo automatismo con MCP sono necessari:

- 2 punti di ingresso collegati ai testimoni dei teleruttori di ognuna delle 2 pompe (normalmente aperti)
- 2 punti di ingresso collegati ai contatti di anomalia di ognuna delle 2 pompe (normalmente chiusi)
- 2 punti di ingresso collegati alle posizioni AUTO e MAN del commutatore che controlla la pompa 1
- 2 punti di ingresso collegati alle posizioni AUTO e MAN del commutatore che controlla la pompa 2
- 2 punti di uscita per il comando dei teleruttori di ognuna delle 2 pompe
- 1 punto di uscita per l'attivazione dell' allarme

Le equazioni che realizzano l'automatismo sopra descritto sono qui di seguito riportate ipotizzando indirizzi e punti a titolo esemplificativo. Ad ognuno di questi indirizzi e punti sara' sufficiente sostituire quelli dell'impianto reale per ottenere le equazioni volute.


```

////////////////////////////////////
// SEQUENZA 2 POMPE GEMELLARI //////////////////////////////////////
////////////////////////////////////
// PUNTI DI INGRESSO E USCITA UTILIZZATI E LORO SIGNIFICATO:
// V5 = CLOCK SUL QUALE E' PRESENTE 1 IMPULSO ALLA SETTIMANA
// I1.1 = TESTIMONE P1
// I1.2 = TESTIMONE P2
// I1.5 = ANOMALIA P1
// I1.6 = ANOMALIA P2
// I2.1 = P1 AUTO
// I2.2 = P1 MAN
// I2.3 = P2 AUTO
// I2.4 = P2 MAN
// O1.1 = P1
// O1.2 = P2
// O1.8 = ALLARME

V5 = CLOCK(LUN:00:00,LUN:00:01) // 1 IMPULSO DA 1 MINUTO
// ALLA SETTIMANA

V1 = C = 1,2 UV5 // VIRTUALE PER SCAMBIO

O1.1 = I2.1 & I1.5 & ( \ // BLOCCO P1 IN AUTO:
                        !V1 & !I2.4 | \ // ON DA CLOCK
                        !I1.6 | \ // ON DA ANOM. P2
                        !I2.3 & !I2.4 \ // ON SE P2 SU 0
                        ) | \ // P1 IN MANUALE
      I2.2 & I1.5

O1.2 = I2.3 & I1.6 & ( \ // BLOCCO P2 IN AUTO:
                        V1 & !I2.2 | \ // ON DA CLOCK
                        !I1.5 | \ // ON DA ANOM. P1
                        !I2.1 & !I2.2 \ // ON SE P1 SU 0
                        ) | \ // P2 IN MANUALE
      I2.4 & I1.6

O1.8 = (!I1.1 & !I1.2) | \ // USCITA ALLARME: ATTIVO
      (!I1.2 & !I1.1) // QUANDO AMBEDUE LE POMPE
// SONO SPENTE

```

Controllo del livello di un serbatoio con due pompe a funzionamento alternato

```

////////////////////////////////////
//////////////////////////////////// SEQUENZA 2 POMPE - 2 LIVELLI //////////////////////////////////////
////////////////////////////////////
// DESCRIZIONE DI FUNZIONAMENTO:
// SI VUOLE CONTROLLARE IL LIVELLO DI UN SERBATOIO MEDIANTE 2 POMPE
// E 2 LIVELLOSTATI. OGNI VOLTA CHE IL LIVELLO SCENDE SOTTO IL LIVELLO 1
// SI DEVE AVVIARE UNA DELLE DUE POMPE; PER OGNI CHIAMATA DEL LIVELLO 1
// SI DEVE AVERE LO SCAMBIO DELLE DUE POMPE, IN MODO DA AVERE UNA UGUALE
// USURA DELLE STESSE.
// SE NONOSTANTE L' AVVIAMENTO DI UNA POMPA IL LIVELLO SCENDE ULTERIORMENTE
// SINO SOTTO IL LIVELLO 2, VIENE AVVIATA ANCHE LA SECONDA POMPA.
// INOLTRE, SE LA POMPA CHE STA FUNZIONANDO VA IN AVARIA (ES. TERMICA
// O MARCIA A VUOTO), DEVE PARTIRE LA SECONDA POMPA.
// IN QUESTA APPLICAZIONE SI E' INOLTRE ASSUNTO DI RITARDARE (AD ES. DI
// 1 SECONDO) LA RILEVAZIONE DEL LIVELLO IN MODO DA EVITARE FALSE PARTENZE
// DOVUTE AD ESEMPIO AL MOVIMENTO DEL LIQUIDO NEL SERBATOIO. PER CAMBIARE
// QUESTI RITARDI E' SUFFICIENTE MODIFICARE I RELATIVI PARAMETRI NELLE DUE
// FUNZIONI TIMER. NELLE EQUAZIONI CHE SEGUONO SI ASSUME, A TITOLO DI ESEMPIO:
// O1.1 = USCITA POMPA 1
// O1.2 = USCITA POMPA 2
// I1.1 = TERMICO POMPA 1 (CHIUDE IN CASO DI ANOMALIA)
// I1.2 = TERMICO POMPA 2 (CHIUDE IN CASO DI ANOMALIA)
// I1.3 = MARCIA A VUOTO POMPA 1 (CHIUDE IN CASO DI ANOMALIA)
// I1.4 = MARCIA A VUOTO POMPA 2 (CHIUDE IN CASO DI ANOMALIA)
// I1.5 = INGRESSO LIVELLOSTATO 1 (APRE SOPRA MAX, CHIUDE SOTTO MIN)
// I1.6 = INGRESSO LIVELLOSTATO 2 (APRE SOPRA MAX, CHIUDE SOTTO MIN)
// I2.1 = INGRESSO FUNZIONAMENTO MANUALE
// I2.2 = INGRESSO FUNZIONAMENTO AUTOMATICO
// I2.3 = INGRESSO COMANDO MANUALE POMPA 1
// I2.4 = INGRESSO COMANDO MANUALE POMPA 2
// LE EQUAZIONI NECESSARIE AL CONTROLLO DELLE DUE POMPE SONO LE SEGUENTI:

V100 = C > 1,2 UI1.5          // VIRTUALE PER SCAMBIO P1 <-> P2
V1 = I1.1 | I1.3             // SOMMATORIA ANOMALIE POMPA 1
V2 = I1.2 | I1.4             // SOMMATORIA ANOMALIE POMPA 2
V151 = TIMER(I1.5,10,10)     // RITARDO L1MIN
V153 = TIMER(I1.6,10,10)     // RITARDO L2MIN

// MARCIA P1
O1.1 = I2.2 & \              // SE AUTO &
      (!V1 & \               // (NO ANOM1 &
      V151 & V100 | \        // L1MIN & SCAMBIO |
      V153 | \               // L2MIN |
      V151 & !V100 & V2) | \ // L1MIN & !SCAMBIO & ANOM2) |
      I2.1 & !V1 & \        // SE MAN. & NO ANOM1 &
      I2.3                   // COMANDO MARCIA 1

// MARCIA P2
O1.2 = I2.2 & \              // SE AUTO &
      (!V2 & \               // (NO ANOM2 &
      V151 & !V100 | \        // L1MIN & !SCAMBIO |
      V153 | \               // L2MIN |
      V151 & V100 & V1) | \ // L1MIN & SCAMBIO & ANOM1) |
      I2.1 & !V2 & \        // SE MAN. & NO ANOM2 &
      I2.4                   // COMANDO MARCIA 2

```

**Gestione 2 pompe gemellari con pressostato mediante MCP
(Versione=<12)**

```

////////////////////////////////////
//////// 2 POMPE GEMELLARI COMANDATE DA PRESSOSTATO //////////
////////////////////////////////////
// FUNZIONAMENTO: SI VUOLE CONTROLLARE IL FUZIONAMENTO DI 2 POMPE MEDIANTE UN
// PRESSOSTATO DI MINIMA SECONDO LA SEGUENTE DESCRIZIONE:
// SE LA PRESS. E' MINORE DELLA SOGLIA PARTE LA POMPA 1; QUANDO LA PRESSIONE
// TORNA SOPRA SOGLIA LA POMPA 1 VIENE SPENTA. SE LA PRESSIONE TORNA SOTTO
// SOGLIA PARTE LA POMPA 2, DOPO DI CHE ESSA VIENE SPENTA QUANDO LA PRESSIONE
// TORNA SOPRA SOGLIA E COSI' VIA.
// SE MENTRE E' IN FUNZIONE UNA POMPA LA PRESSIONE NON TORNA SOPRA SOGLIA
// ENTRO UN TEMPO DEFINITO, PARTE ANCHE L' ALTRA POMPA ED ESSE RIMANGONO
// ENTRAMBE IN FUNZIONE FINO A CHE LA PRESSIONE NON TORNA SOPRA SOGLIA.
// IL FUNZIONAMENTO SOPRA DESCRITTO E' CONTROLLATO DA UN SELETTORE AUTO/0/MAN
// COME SEGUE:
// * IN POSIZIONE MAN IL FUNZIONAMENTO E' QUELLO SOPRA DESCRITTO
// * IN POSIZIONE 0 AMBEDUE LE POMPE SONO SPENTE
// * IN POSIZIONE AUTO AMBEDUE LE POMPE SONO ACCESE
// INOLTRE E' PRESENTE UN PRESSOSTATO DI MASSIMA CHE SPENGE IN OGNI CASO LE
// POMPE SE LA PRESSIONE SALE SOPRA IL SUO VALORE DI SOGLIA.
// NELL 'ESEMPIO SEGUENTE SI E' CONSIDERATO:
// I2.1 = FUNZ. AUTO SE CONTATTO CHIUSO
// I2.2 = FUNZ. MAN SE CONTATTO CHIUSO
// I2.3 = PRESSOSTATO DI MINIMA (CHIUDE SE PRESS. < SOGLIA)
// I2.4 = PRESSOSTATO DI MASSIMA (CHIUDE SE PRESS. > SOGLIA)
// O2.1 = QUANDO ATTIVA BYPASSA INVERTER
// O2.2 = USCITA POMPA 1
// O2.3 = USCITA POMPA 2
// O2.4 = QUANDO ATTIVA ABILITA INVERTER

V100 = TI2.3 // VIRTUALE PER SCAMBIO P1 <-> P2
V150 = TIMER(I2.3,30,0) // RITARDO ACCENSIONE SECONDA POMPA ( PR>TIM )
// ATTIVO SE PR. ANCORA SOPRA SOGLIA DOPO x SECONDI

O2.1 = I2.2 // BYPASSA INVERTER SE IN MANUALE

// USCITA P1
O2.2 = I2.2 & !I2.4 & \ // SE MAN & !PR> &
(I2.3 & V100 | \ // (PR< & SCAMBIO |
V150) | \ // PR>TIM) |
I2.1 & !I2.4 // SE AUTO & !PR>

// USCITA P2
O2.3 = I2.2 & !I2.4 & \ // SE MAN & !PR> &
(I2.3 & !V100 | \ // (PR> & !SCAMBIO |
V150) | \ // PR>TIM) |
I2.1 & !I2.4 // SE AUTO & !PR>

O2.4 = TIMER (I2.1,100,0) // ABILITAZIONE INVERTER

```

Conversione da codice binario a uscite dirette

```

////////////////////////////////////
//////// PROGRAMMA PER LA CONVERSIONE DI INGRESSO BINARIO //////////
//////// IN USCITE DIRETTE //////////////////////////////////////////
////////////////////////////////////
//
// DESCRIZIONE:
// UTILIZZANDO LE SCHEDE D2BIN-M E D2BIN-S, E' POSSIBILE UTILIZZARE
// SOLO MODULO DI INGRESSO (MODINP2/A OPPURE MOD8I/A) PER LEGGERE
// SINO AD UN MASSIMO DI 127 INGRESSI DIRETTI. LA SCHEDA D2BIN
// PRESENTA ALLA SUA USCITA IL CODICE BINARIO DELL'INGRESSO ATTIVO;
// NEL CASO IN CUI GLI INGRESSI ATTIVI FOSSERO PIU' DI UNO, LA SCHEDA
// D2BIN METTE IN USCITA I RELATIVI CODICI CON UN PERIODO DI CIRCA 2
// SECONDI. SE L'USCITA DELLA D2BIN-M VIENE CABLATA AD UN MODULO DI
// INGRESSO DELLA SERIE CONTATTO, E' POSSIBILE RICONVERTIRE IL CODICE
// IN UNA INFORMAZIONE DIRETTA; IL PROGRAMMA CHE SEGUE "SVILUPPA" IL
// CODICE PRESENTE SULL'INGRESSO DEL MODULO IN 127 PUNTI VIRTUALI.
// RISULTA CHIARO CHE SE VI SONO PIU' INGRESSI ATTIVI SULLA D2BIN-M
// (O D2BIN-S), I RELATIVI PUNTI VIRTUALI DI MCP VERRANNO ATTIVATI
// COMUNQUE UNO SOLO ALLA VOLTA CON UN PERIODO DI 2 SECONDI COME
// SOPRA RICORDATO.
// CONNESSIONI TRA SCHEDA D2BIN-M E MODULO DI INGRESSO
// (INDIFFERENTEMENTE MODINP2/A O MOD8I/A):
// MORSETTI MODINP          MORSETTI D2BIN-M
//      1 ----- D0
//      2 ----- D1
//      3 ----- D2
//      4 ----- D3
//      5 ----- D4
//      6 ----- D5
//      7 ----- D6
//     14 ----- GND
//
// DATA LA FACILITA' DI COMMITTERE ERRORI DI BATTITURA RICOPIANDO
// IL PROGRAMMA CHE SEGUE, CONSIGLIAMO DI CONTATTARE L'UFFICIO
// TECNICO DUEMMEGI PER AVERE, VIA E-MAIL O TRAMITE DISCHETTO,
// IL PROGRAMMA GIA' SCRITTO E COLLAUDATO. NEL CASO IN CUI STATE
// LEGGENDO QUESTO DOCUMENTO IN FORMATO ACROBAT® , E' POSSIBILE
// ESEGUIRE UN "COPIA - INCOLLA".
//
V1 = !I1.7 & !I1.6 & !I1.5 & !I1.4 & !I1.3 & !I1.2 & I1.1
V2 = !I1.7 & !I1.6 & !I1.5 & !I1.4 & !I1.3 & I1.2 & !I1.1
V3 = !I1.7 & !I1.6 & !I1.5 & !I1.4 & !I1.3 & I1.2 & I1.1
V4 = !I1.7 & !I1.6 & !I1.5 & !I1.4 & I1.3 & !I1.2 & !I1.1
V5 = !I1.7 & !I1.6 & !I1.5 & !I1.4 & I1.3 & !I1.2 & I1.1
V6 = !I1.7 & !I1.6 & !I1.5 & !I1.4 & I1.3 & I1.2 & !I1.1
V7 = !I1.7 & !I1.6 & !I1.5 & !I1.4 & I1.3 & I1.2 & I1.1
V8 = !I1.7 & !I1.6 & !I1.5 & I1.4 & !I1.3 & !I1.2 & !I1.1
V9 = !I1.7 & !I1.6 & !I1.5 & I1.4 & !I1.3 & !I1.2 & I1.1
V10= !I1.7 & !I1.6 & !I1.5 & I1.4 & !I1.3 & I1.2 & !I1.1
V11= !I1.7 & !I1.6 & !I1.5 & I1.4 & !I1.3 & I1.2 & I1.1
V12= !I1.7 & !I1.6 & !I1.5 & I1.4 & I1.3 & !I1.2 & !I1.1
V13= !I1.7 & !I1.6 & !I1.5 & I1.4 & I1.3 & !I1.2 & I1.1
V14= !I1.7 & !I1.6 & !I1.5 & I1.4 & I1.3 & I1.2 & !I1.1
V15= !I1.7 & !I1.6 & !I1.5 & I1.4 & I1.3 & I1.2 & I1.1
V16= !I1.7 & !I1.6 & I1.5 & !I1.4 & !I1.3 & !I1.2 & !I1.1
V17= !I1.7 & !I1.6 & I1.5 & !I1.4 & !I1.3 & !I1.2 & I1.1

```


Pulsanti passo-passo locali e un pulsante generale di ON-OFF

```

// DATI 10 PUNTI LUCE, OGNUNO E' COMANDATO MEDIANTE
// PULSANTE LOCALE CON FUNZIONE PASSO-PASSO. UN PULSANTE
// GENERALE DEVE ACCENDERE TUTTO SE LE LUCI SONO TUTTE
// SPENTE E DEVE SPEGNERE TUTTO SE TUTTE ACCESE.
// SE LE LUCI SONO DISCORDI (UN PO' ACCESE E UN PO'
// SPENTE, ALLORA IL PULSANTE GENERALE FORZA
// L'ACCENSIONE PER RIALLINEARE TUTTO.

#define %P1% I1.1
#define %P2% I1.2
#define %P3% I1.3
#define %P4% I1.4
#define %P5% I1.5
#define %P6% I1.6
#define %P7% I1.7
#define %P8% I1.8
#define %P9% I2.1
#define %P10% I2.2
#define %PG% I2.8

V101 = ( V1 & V2 & V3 & V4 & V5 & \ // V101=1 SE LUCI CONCORDI
        V6 & V7 & V8 & V9 & V10) | \
        (!V1 & !V2 & !V3 & !V4 & !V5 & \
         !V6 & !V7 & !V8 & !V9 & !V10)

V102 = %PG% & V101
V103 = %PG% & !V101 | V103 & %PG%

// NELLE SEGUENTI RIGHE OGNI USCITA VIENE COMMUTATA (TOGGLE) DA PULSANTE
// LOCALE, OPPURE DA PULSANTE CUMULATIVO SE LE USCITE SONO CONCORDI.
// INOLTRE, SE LE USCITE SONO DISCORDI, OGNI USCITA VIENE FORZATA ACCESA
// DALLO STESSO PULSANTE CUMULATIVO.
V1 = T%P1% | TV102 | SV103 | R!V998
V2 = T%P2% | TV102 | SV103 | R!V998
V3 = T%P3% | TV102 | SV103 | R!V998
V4 = T%P4% | TV102 | SV103 | R!V998
V5 = T%P5% | TV102 | SV103 | R!V998
V6 = T%P6% | TV102 | SV103 | R!V998
V7 = T%P7% | TV102 | SV103 | R!V998
V8 = T%P8% | TV102 | SV103 | R!V998
V9 = T%P9% | TV102 | SV103 | R!V998
V10 = T%P10% | TV102 | SV103 | R!V998

// USCITE LUCI
O1.1 = V1
O1.2 = V2
O1.3 = V3
O1.4 = V4
O1.5 = V5
O1.6 = V6
O1.7 = V7
O1.8 = V8
O2.1 = V9
O2.2 = V10

```

Pulsanti passo-passo locali, pulsanti di gruppo e un pulsante generale di ON-OFF

```
// DATI 4 PUNTI LUCE, OGNUNO E' COMANDATO MEDIANTE
// PULSANTE LOCALE CON FUNZIONE PASSO-PASSO.
// DUE PULSANTI "DI GRUPPO" COMANDANO CIASCUNO
// LE PRIME DUE LAMPADE E LE ALTRE DUE IN ACCENSIONE
// SE SONO SPENTE E IN SPEGNIMENTO SE SONO ACCESE; NEL
// CASO IN CUI FOSSERO DISCORDI, LE ACCENDE PER RIALLINEARE.
// UN ULTERIORE PULSANTE GENERALE DEVE ACCENDERE TUTTO
// SE LE LUCI SONO TUTTE SPENTE E DEVE SPEGNERE TUTTO
// SE TUTTE ACCESE. SE LE LUCI SONO DISCORDI (UN PO'
// ACCESE E UN PO' SPENTE, ALLORA IL PULSANTE GENERALE
// FORZA L'ACCENSIONE PER RIALLINEARE TUTTO.
// RIASSUMENDO:
// OGNI USCITA VIENE COMMUTATA (TOGGLE) DA PULSANTE LOCALE,
// OPPURE DA PULSANTE CUMULATIVO (DI GRUPPO O GENERALE) SE LE
// USCITE SONO CONCORDI. INOLTRE, SE LE USCITE SONO DISCORDI,
// OGNI USCITA VIENE FORZATA ACCESA DALLO STESSO PULSANTE
// CUMULATIVO (DI GRUPPO O GENERALE).

// DEFINIZIONI PULSANTI
// PULSANTI LOCALI
#define %P1% I1.1
#define %P2% I1.2
#define %P3% I1.3
#define %P4% I1.4

#define %PG1% I1.5
#define %PG2% I1.6
#define %PG% I1.7

////////////////////////////////////
// DEFINIZIONE VIRTUALI PER COMANDO GENERALE: //////////////////////////////////
////////////////////////////////////
V101 = ( V1 & V2 & V3 & V4 ) | \ // V101=1 SE TUTTE LE
      (!V1 & !V2 & !V3 & !V4 ) // LUCI SONO CONCORDI

V102 = %PG% & V101 // V102 ATTIVA CON PG PREMUTO
      // E LUCI CONCORDI

V103 = %PG% & !V101 | V103 & %PG% // V103 ATTIVA CON PG PREMUTO
      // E LUCI DISCORDI

////////////////////////////////////
// GRUPPO 1 //////////////////////////////////
////////////////////////////////////
// DEFINIZIONE VIRTUALI PER COMANDO GRUPPO 1:
V104 = ( V1 & V2 ) | \ // V104=1 SE TUTTE LE DEL GRUPPO
      (!V1 & !V2 ) // SONO CONCORDI
V105 = %PG1% & V104 // V105 ATTIVA CON PG1 PREMUTO
      // E LUCI DEL GRUPPO CONCORDI
V106 = %PG1% & !V104 | V106 & %PG1% // V106 ATTIVA CON PG1 PREMUTO
      // E LUCI DEL GRUPPO DISCORDI

V1 = T%P1% | TV105 | TV102 | SV106 | SV103 | R!V998
V2 = T%P2% | TV105 | TV102 | SV106 | SV103 | R!V998
```



```
// USCITE LUCI GRUPPO 1
O1.1 = V1
O1.2 = V2

////////////////////////////////////
////////// GRUPPO 2 //////////////////////////////////////
////////////////////////////////////
// DEFINIZIONE VIRTUALI PER COMANDO GRUPPO 2:
V107 = ( V3 & V4 ) | \ // V107=1 SE TUTTE LE DEL GRUPPO
      (!V3 & !V4 ) // SONO CONCORDI
V108 = %PG2% & V107 // V108 ATTIVA CON PG2 PREMUTO
// E LUCI DEL GRUPPO CONCORDI
V109 = %PG2% & !V107 | V109 & %PG2% // V109 ATTIVA CON PG2 PREMUTO
// E LUCI DEL GRUPPO DISCORDI
V3 = T%P3% | TV108 | TV102 | SV109 | SV103 | R!V998
V4 = T%P4% | TV108 | TV102 | SV109 | SV103 | R!V998

// USCITE LUCI GRUPPO 1
O1.3 = V3
O1.4 = V4
```

Guida alla configurazione dei modem in un sistema PC – MCP collegati via linea telefonica

La presente nota applicativa si riferisce ad un sistema PC – MODEM che comunica con un sistema MODEM–MCP (o MCP Plus) per mezzo di una linea telefonica pubblica; i più frequenti “problemi” che si incontrano quando si vuole mettere in servizio tale sistema riguardano quasi esclusivamente la configurazione dei due modem (uno connesso al PC ed uno connesso a MCP).

ATTENZIONE: questa nota applicativa si riferisce alla versione 4.2 (o superiori) di MCP, a MCP Plus ed alla release 2.21 (o superiori) di MCPTOOLS. Quanto segue si riferisce inoltre a modem DIGICOM mod. TIZIANO 33.6 oppure mod. LEONARDO 56 oppure mod. BOTTICELLI; dato che tra modem di diversi costruttori ci possono essere alcune differenze nelle impostazioni, non è detto che quanto segue funzioni con qualsiasi modem.

In questa nota applicativa verranno elencate, passo dopo passo, le procedure da seguire per effettuare un collegamento tra PC e MCP remoto e viceversa.

Come noto, è possibile programmare MCP in modo che al verificarsi di uno o più eventi (allarmi o altro), esso generi una chiamata via modem; per far ciò si deve includere nel programma caricato in MCP il blocco MODEM (vedere manuale per dettagli).

Nel blocco MODEM è possibile includere sino a 3 stringhe di comandi AT per configurare automaticamente il modem alla accensione del modulo MCP (solo per MCP versione 4.2 o superiore); la lunghezza massima di ognuna di queste stringhe è di 17 caratteri. Le stringhe di configurazione devono essere posizionate, all'interno del blocco MODEM, prima dei 4 numeri di telefono.

La configurazione del modem collegato ad MCP, che è il primo passo da fare, sarà quindi eseguita direttamente da MCP stesso alla sua accensione a patto di aggiungere al blocco MODEM, prima dei numeri di telefono, le stringhe di configurazione come descritto nel seguito.

Passo 1:

Aggiungere, mediante MCPTOOLS, al programma di MCP il blocco MODEM (o includere le stringhe di configurazione se il blocco MODEM esiste già) come segue:

```
MODEM (           \      // Inizio blocco
  AT&FOE0         \      // Stringa 1 di configurazione MODEM
  AT-K0X3&K0&D0  \      // Stringa 2 di configurazione MODEM
  ATT&WZ         \      // Stringa 3 di configurazione MODEM
  TEL1 = ...     \      // Numero telefonico 1
  TEL2 = ...     \      // Numero telefonico 2
  TEL3 = ...     \      // Numero telefonico 3
  TEL4 = ...     \      // Numero telefonico 4
  V...          \      // Allarme 1
  ...           \      // eccetera...
)                \      // Fine blocco
```

ATTENZIONE: per il modem DIGICOM Botticelli WEB la seconda stringa di configurazione deve essere: **ATX3&K0&D0**. In tutte le stringhe di configurazione, assicurarsi di scrivere 0 (zero) e non il carattere O.

Se non si vuole che MCP chiami al verificarsi di un evento, includere comunque il blocco modem nel programma fermandosi alla definizione dei numeri di telefono (che possono essere fittizi ma obbligatori); in altre parole il blocco MODEM sarà esattamente il seguente:

```

MODEM (           \      // Inizio blocco
  AT&FOE0         \      // Stringa 1 di configurazione MODEM
  AT-K0X3&K0&D0  \      // Stringa 2 di configurazione MODEM
  ATT&WZ         \      // Stringa 3 di configurazione MODEM
  TEL1 = 123     \      // Numero telefonico 1
  TEL2 = 123     \      // Numero telefonico 2
  TEL3 = 123     \      // Numero telefonico 3
  TEL4 = 123     \      // Numero telefonico 4
)                 \      // Fine blocco
    
```

ATTENZIONE: per il modem DIGICOM Botticelli WEB la seconda stringa di configurazione deve essere: ATX3&K0&D0. nel caso in cui la rete telefonica e/o il centralino accetti solo la selezione a impulsi, anziché a toni, modificare la terza stringa di configurazione come segue:

```
ATP&WZ
```

Passo 2:


Compilare e trasferire a MCP (mediante MCPTOOLS) il programma completo di blocco MODEM.

Passo 3:

Spegnere l'alimentazione di MCP e collegarlo al MODEM mediante apposito cavo seriale. Si ricorda che il collegamento seriale RS232 tra modulo MCP e modem è a 3 fili (TX, RX e comune) come indicato nel seguente schema:

Connettore 25 poli maschio


Connettore 9 poli maschio



Se il modem ha un connettore a 9 poli (come il DIGICOM Botticelli WEB), il cavo deve essere come segue:

Connettore 9 poli maschio

Connettore 9 poli maschio



ATTENZIONE: il cavo fornito in dotazione con il modem non è adatto al collegamento con MCP.

Nota: il cavalletto tra i pin 4 e 6 sul connettore lato MCP è indispensabile per il corretto funzionamento con MCP Plus.

Passo 4:


Accendere il MODEM e, dopo circa 10 secondi, accendere MCP; il MODEM verrà automaticamente configurato da MCP nel giro di circa 15 secondi (durante questo tempo i LED TX e RX su MCP emettono brevi lampeggi). Per questa operazione non è necessario che il MODEM sia collegato alla rete telefonica e non è necessario che MCP sia collegato ai moduli della famiglia Contatto).

Passo 5:


Collegare il PC con l'altro MODEM. Lanciare il programma di supporto MCPTOOLS (release 2.21 o superiore).

Passo 6:

Dal menu di MCPTOOLS selezionare Connect – MODEM:




Apparirà la seguente finestra:



Premere il button Comm e selezionare la porta COM in uso sul PC e la velocità (quest'ultima è generalmente 19200); la funzione Auto Detect, nel caso del modem, non è disponibile. Premere il button Close.

Passo 7:

Premere il button Config; apparirà la seguente finestra:




Passo 8:

Nelle 3 caselle inserire le seguenti stringhe:

AT&F0E0
 AT-K0X3&K0&D2
 AT&WZ

ATTENZIONE: per il modem DIGICOM Botticelli WEB la seconda stringa di configurazione deve essere: **ATX3&K0&D0**. In tutte le stringhe di configurazione, assicurarsi di scrivere 0 (zero) e non il carattere O. Alla fine la finestra apparirà come segue:



Passo 9:

Premere il button Save per memorizzare su disco le stringhe di configurazione.

Passo 10:


Premere il button Send per trasferire la configurazione al MODEM (questo processo dura circa 10 secondi).

Passo 11:

Premere il button Close per tornare alla finestra "MCP Connection by MODEM". A questo punto anche il secondo MODEM è configurato.

Nota: a qualsiasi successiva apertura della finestra MODEM CONFIGURATION, le 3 stringhe sopra esposte saranno già presenti in quanto salvate in un file di configurazione su disco. Per trasferire le stringhe al modem sarà dunque sufficiente premere il button Send presente nella stessa finestra (in altre parole saltare i passi 8 e 9).

Passo 12:



Nella finestra qui sopra riportata, premere il button Reset Modem: se nella sottofinestra Status apparirà OK, significa che le operazioni di configurazione sono state eseguite correttamente.

Passo 13:

A questo punto è possibile effettuare un test di collegamento con MCP remoto mediante linea telefonica:

- comporre il numero da chiamare nella casella Telephone number e premere il pulsante Dial
- se la comunicazione avrà esito positivo, nella casella Status apparirà un messaggio confermando l'avvenuta connessione (CONNECT ...) oppure un messaggio d'errore in caso di fallimento (NO CARRIER, BUSY, ...)
- è possibile a questo punto, se la connessione è stata effettuata, conoscere l'identificativo della stazione chiamata (pulsante MCP ID) oppure leggere la lista degli allarmi intervenuti e azzerarla (pulsanti Read e Clear).

Per chiudere la comunicazione utilizzare il pulsante Hang Up.

Lasciando aperta la comunicazione e chiudendo solamente la finestra con il pulsante Close è possibile utilizzare tutte le funzioni di MCP-Tools, vale a dire programmazione, visualizzazione mappa, ecc., come se il modulo MCP fosse collegato direttamente al PC.

Il pulsante Answer consente di rispondere alle chiamate entranti e quindi di collegarsi all'impianto remoto che ha effettuato la chiamata.

ATTENZIONE: la procedura di configurazione dei modem descritta in questa nota applicativa va seguita, per quanto riguarda il MODEM lato MCP, anche nel caso in cui si utilizzi poi il programma di supervisione TRUE VISION.

Gestione di sequenze allarmi con MCP e DISPBUS

In questo paragrafo viene mostrato, come implementare una sequenza allarmi totalmente gestita dal modulo controllore **Contatto MCP** abbinato ad un visualizzatore **DISPBUS**.

Il DISPBUS nell'esempio che segue **deve** essere configurato come **NOMEM** e **REMOTE**; è inoltre stato assegnato l'indirizzo base 003 (ma potrebbe essere anche un altro indirizzo).

Descrizione di funzionamento:

Dati 8 ingressi (I1.1 + I1.8) si vuole che alla chiusura di ognuno di essi si verifichi quanto segue:

- viene visualizzato sul DISPBUS il messaggio relativo all'evento occorso
- l'evento viene memorizzato, per cui anche se l'ingresso torna a riposo, la condizione di allarme rimane attiva
- le uscite sirena e lampeggiatore del DISPBUS vengono attivate

Si vuole tacitare la sirena premendo il tasto ACK sul DISPBUS; la pressione di questo tasto non deve invece avere effetto alcuno sull'uscita lampeggiatore e sulla visualizzazione. La pressione del tasto RESET sul DISPBUS deve invece "riarmare" il sistema, vale a dire deve rimuovere dalla memoria gli allarmi il cui ingresso è tornato a riposo e, se non vi sono più allarmi presente, deve spegnere l'uscita lampeggiatore.

Dopo la tacitazione della sirena, ogni nuovo allarme (vale a dire non ancora memorizzato) deve far ripartire la sirena stessa.

Inoltre il tasto RESET non deve avere alcun effetto se prima la sirena non è stata tacitata.

Il programma per il modulo **Contatto MCP** che implementa il funzionamento appena descritto è il seguente:

```

////////////////////////////////////
// PROGRAMMA GESTIONE ALLARMI CON DISPBUS //
////////////////////////////////////
// DEFINIZIONE MEMORIE ALLARMI: OGNI CELLA DI MEMORIA E' SETTATA DAL RELATIVO
// INGRESSO (DA I1.1 A I1.8) E RESETTATA DAL PULSANTE DI RESET SUL DISPBUS I3.2)
// A PATTO CHE LA SIRENA (O4.7=V202) SIA STATA TACITATA E CHE IL RELATIVO
// INGRESSO SIA TORNATO A RIPOSO

V1 = SI1.1 & (RV203 | I1.1)
V2 = SI1.2 & (RV203 | I1.2)
V3 = SI1.3 & (RV203 | I1.3)
V4 = SI1.4 & (RV203 | I1.4)
V5 = SI1.5 & (RV203 | I1.5)
V6 = SI1.6 & (RV203 | I1.6)
V7 = SI1.7 & (RV203 | I1.7)
V8 = SI1.8 & (RV203 | I1.8)

// USCITA SIRENA (O4.7): OGNI NUOVO ALLARME FA PARTIRE LA SIRENA; LA TACITAZIONE
// AVVIENE MEDIANTE IL PULSANTE DI ACK SUL DISPBUS (I3.1)

V201 = TV1 | TV2 | TV3 | TV4 | \
      TV5 | TV6 | TV7 | TV8 | \
      RI3.1
V202 = SV201 & RI3.1 & R!V998
O4.7 = V202
    
```

```
// VIRTUALE PER IL RESET DELLE CELLE DI MEMORIA DEGLI ALLARMI: V203 DIVENTA
// ATTIVA PREMENDO IL PULSANTE DI RESET (I3.2) A PATTO CHE LA SIRENA
// (O4.7=V202) SIA SPENTA
```

```
V203 = !V202 & I3.2
```

```
// USCITA LAMPEGGIATORE (O4.8), DEFINITO COME OR DELLE CELLE DI ALLARME;
// IL LAMPEGGIATORE RIMANE ATTIVO SINO A CHE NON SONO STATI RIMOSSI TUTTI
// GLI ALLARMI E RESETTATO IL SISTEMA
```

```
V204 = V1 | V2 | V3 | V4 | \
        V5 | V6 | V7 | V8
O4.8 = V204
```

```
// DEFINIZIONE BLOCCO DI USCITA BINARIO
```

```
BINARY 3 ( \
    B001 = V1    \
    B002 = V2    \
    B003 = V3    \
    B004 = V4    \
    B005 = V5    \
    B006 = V6    \
    B007 = V7    \
    B008 = V8    \
)
```

Questo programma può essere facilmente modificato per essere adattato ai vari casi reali. Per quanto riguarda i dettagli sulla programmazione dei messaggi nel visualizzatore DISPBUS e relative configurazioni, fare riferimento al manuale dello stesso.

Termoregolazione ambientale mediante modulo CLIMA e MCP

Il modulo **Contatto** CLIMA consente di implementare un impianto di termoregolazione di un ambiente; questo modulo mette a disposizione essenzialmente 2 funzioni:

1. misura della temperatura di una locale (nel campo tra 0 e 40 gradi Centigradi)
2. funzione termostato, il cui set point può essere variato localmente mediante un potenziometro (rotella) posizionata sul modulo stesso (nel campo tra 15 e 30 gradi Centigradi)

Il modulo CLIMA si comporta come un modulo di ingresso e il codice binario restituito a MCP contiene 2 informazioni:

- l'intero codice è la temperatura misurata in formato binario a 8 bit, nel campo tra 0 e 40 gradi; in altre parole, quando il byte restituito dal modulo CLIMA è zero, allora la temperatura è zero (o inferiore), mentre se il codice è 255, allora la temperatura è 40 gradi (o maggiore)
- il bit meno significativo (vale a dire il punto 1 del modulo di ingresso CLIMA) ha inoltre il seguente significato:
 - ✓ esso è uguale a zero per temperature ambiente minori del set point definito dal potenziometro posto sul modulo
 - ✓ esso è uguale a 1 per temperature ambiente superiori del set point definito dal potenziometro posto sul modulo

Inoltre il modulo CLIMA aggiunge automaticamente una isteresi di 1 grado attorno al set point "locale", in modo da evitare oscillazioni quando la temperatura è vicina al punto di commutazione; riassumendo, detto D0 il bit meno significativo del codice binario restituito dal modulo CLIMA, Ta la temperatura ambiente e SP il set point locale impostato da potenziometro, si ha:

D0 = 1 se Ta > SP + 0.5 gradi

D0 = 0 se Ta < SP - 0.5 gradi

Nota: la posizione del potenziometro, e quindi il set point locale, NON possono essere letti in alcun modo da MCP.

Il modulo **Contatto** CLIMA consente quindi di implementare un impianto di termoregolazione ambiente, permettendo di:

- dare la possibilità all'occupante della camera ove il modulo è installato di regolare il set point di temperatura PER QUELLA CAMERA secondo suo piacimento (funzione termostato)
- rendere possibile la lettura delle temperature REALI delle varie camere ove sono installati i moduli CLIMA (ad esempio da un sistema di supervisione locale o collegato via modem) (funzione misura di temperatura)

Come detto, la temperatura della camera viene comunicata dal modulo CLIMA a MCP sotto forma di codice binario a 8 bit nel campo di misura da 0 a 40 gradi Centigradi; la relazione tra la temperatura misurata ed il codice binario è data dalla seguente formula:

$$T = \frac{n \times 40}{255}$$

dove T è la temperatura e n è il codice binario restituito dal modulo CLIMA.

Viceversa, la relazione tra codice binario e temperatura è il seguente:

$$n = \frac{255 \times T}{40}$$

La regolazione di temperatura di una camera può essere eseguita rispetto al set point impostato mediante la manopola a bordo del modulo CLIMA oppure rispetto ad un set point programmato in MCP oppure ancora rispetto ad un set point impostato da supervisore.

Ad esempio, detto 1 l'indirizzo del modulo di ingresso CLIMA e detta 01.1 una uscita che, ad esempio, controlla il fan-coil della camera, si potrebbe avere:

```
V1 = A1 < 127,3          // virtuale attiva quando la temperatura e' < 20 gradi
                          // (con 3 step di isteresi, vale a dire circa 0.9 gradi)
```

Detta V200 una virtuale controllata da PC oppure da commutatore (ad esempio situato alla reception) e definito che se V200 = 0 il controllo deve avvenire in remoto, mentre se V200 = 1 il controllo deve avvenire rispetto al set point impostato da manopola su modulo CLIMA, si può scrivere:

```
01.1 = V200 & !I1.1 | !V200 & V1    // uscita ON quando V200 e' attivata (modo
                                     // locale) e I1.1 e' zero (che vuol dire
                                     // quando la temp. della camera e' minore
                                     // del set point impostato da manopola),
                                     // oppure quando V200 non e' attivata (modo
                                     // remoto) e V1 e' 1 (temperatura della
                                     // camera minore di 20 gradi).
```

Nell'esempio appena riportato, quindi, il controllo della temperatura avviene rispetto al set point locale, se V200 = 1, oppure rispetto al set point fisso programmato in MCP, se V200 = 0.

E' inoltre possibile, a patto di utilizzare un MCP versione 4.x o superiore, rendere variabile la soglia mediante l'uso di un registro contatore al posto del valore fisso nella definizione della V1 (come descritto sul manuale di MCP versione 4.x).

Per implementare un sistema di supervisione che visualizzi la temperatura delle camere, si deve semplicemente leggere, mediante un opportuno software di supervisione (es. True Vision), la zona di memoria RAM relativa agli ingressi (esattamente come si fa normalmente per gli altri tipi di moduli di ingresso della famiglia **Contatto**); il codice letto in questo modo va elaborato secondo le formule riportate sopra per avere la misura voluta in gradi Centigradi.

Gestione di una uscita analogica per controllare un dimmer

```
// Nota: il programma che segue richiede MCP versione 4.x o superiore
// Scopo di questo programma e' quello di gestire una uscita analogica 0:10V
// che a sua volta controlla un dimmer per la regolazione della luminosita'
// di una lampada.
// Il sistema e' composto da:
// - due pulsanti per il comando UP e DOWN (incrementa - diminuisci)
// - due pulsanti ognuno dei quali forza un valore predefinito di luminosita'
// Premendo e mantenendo premuto il pulsante di UP o DOWN, la luminosita'
// aumenta o diminuisce in modo continuo, grazie alla definizione di una base
// dei tempi.
// Per implementare tale funzione si utilizza un contatore; quando il contatore
// contiene il valore 0 la lampada e' spenta (0 Volt su uscita analogica),
// quando contiene il valore 255 la lampada e' accesa al massimo (10 Volt su
// uscita analogica).
```

```
V990 = TIMER (V991,2,2)          // DEFINIZIONE TEMPO DI AUTOINCREMENTO
                                // (0.4 SECONDI)
```

```
V991 = !V990 & (I1.1 | I1.2)    // I1.1 E I1.2 SONO I PULSANTI UP E DOWN
```

```
V1 = I1.1 & V991                // VIRTUALE PER COMANDO UP
V2 = I1.2 & V991                // VIRTUALE PER COMANDO DOWN
```

```
V3 = C0=1 U[15]V1 & D[15]V2 & \ // UP & DOWN (A PASSI DI 15 STEP)
      P[100]I1.3 & P[200]I1.4 & \ // 2 PRESET
      O1                          // TRASFERIMENTO SU USCITA ANALOGICA 1
```

Esempi di messaggi per la realizzazione di un driver di comunicazione con MCP (protocollo FXP)

Formato del messaggio

La richiesta da inviare a MCP e' costituita da un messaggio formato da:

Indirizzo MCP	Codice	# byte	Dato 1	...	Dato N	Check H	Check L
---------------	--------	--------	--------	-----	--------	---------	---------

dove:

- Indirizzo MCP: 1 byte, indirizzo MCP (station address); se nel messaggio di richiesta si pone l'indirizzo uguale a zero, il messaggio viene accettato qualunque sia l'indirizzo di MCP (0x00 è l'indirizzo jolly, vale quindi qualunque sia l'indirizzo di MCP).
- Codice: 1 byte, codice del messaggio, identifica la richiesta/risposta
- # byte: 1 byte, numero di byte dati che seguono
- Check (H+L) 2 byte, XOR tra la somma dei byte precedenti e 0xFFFF

La risposta da parte di MCP ha lo stesso formato.

MCP risponde ad ogni richiesta con un messaggio simile, dove il campo indirizzo contiene l'indirizzo corrente di MCP (station address); se non è stato programmato alcun indirizzo in MCP, allora il suo campo indirizzo conterrà 0x00.

Nota: l'espressione 0xnn indica il numero nn in notazione esadecimale

Esempio 1: Lettura dello stato degli ingressi

Inviare a MCP il messaggio 0x14, specificando nel campo dati, l'indirizzo di partenza della mappa dei moduli di ingresso e quanti byte si intendono leggere:

Indirizzo di partenza mappa moduli di ingresso: 0x000000 (3 byte)
 Byte da leggere: 128 (1 byte)

Indirizzo MCP	Codice	# byte	Dato 1	Dato 2	Dato 3	Dato 4	Check H	Check L
0x00	0x14	0x04	0x00	0x00	0x00	0x80	0xFF	0x67

MCP risponde con il messaggio contenente 128 byte dati:

Indirizzo MCP	Codice	# byte	Dato 1	Dato 2	...	Dato 128	Check H	Check L
xx	0x14	0x80	0x00	11	...	1127	xx	xx

in cui i 128 byte indicano lo stato dei moduli di ingresso da 0 a 127 (codificati in binario). Ovviamente il modulo 0 non esiste, per cui il primo byte non è significativo.

Ad esempio se Dato 4 è pari a 0x14, che equivale a 00010100 in binario, significa che gli ingressi 3 e 5 del modulo di indirizzo 4 sono attivi.

Nota: xx indica un dato variabile in funzione del messaggio.

Esempio 2: Lettura dello stato delle uscite

Inviare a MCP il messaggio 0x14, specificando nel campo dati, l'indirizzo di partenza della mappa dei moduli di uscita e quanti byte si intendono leggere:

Indirizzo di partenza mappa moduli di ingresso: 0x000080 (3 byte)
 Byte da leggere: 128 (1 byte)

Indirizzo MCP	Codice	# byte	Dato 1	Dato 2	Dato 3	Dato 4	Check H	Check L
0x00	0x14	0x04	0x00	0x00	0x80	0x80	0xFE	0xE7

La risposta è analoga a quella già vista prima per gli ingressi.

Esempio 3: Lettura dello stato degli ingressi e delle uscite mediante un solo messaggio

Essendo la mappa delle uscite contigua all mappa degli ingressi è possibile leggere tutti gli stati dei 127 moduli di ingresso e dei 127 moduli di uscita con un solo messaggio.

Ciò ottimizza i tempi di richiesta quando interessano gli stati di entrambi i tipi di moduli (ingressi ed uscite).

Inviare a MCP il messaggio 0x14, specificando nel campo dati, l'indirizzo di partenza della mappa dei moduli e quanti byte si intendono leggere:

Indirizzo di partenza mappa moduli: 0x000000 (3 byte)
 Byte da leggere: 256 (1 byte)

Indirizzo MCP	Codice	# byte	Dato 1	Dato 2	Dato 3	Dato 4	Check H	Check L
0x00	0x14	0x04	0x00	0x00	0x00	0x00	0xFF	0xE7

Si fa notare che il dato 4 è uguale a 0x00 per indicare una richiesta di 256 byte da leggere; ciò vale solo per questo tipo di messaggio (Codice = 0x14).

MCP risponde con il messaggio contenente 256 byte dati:

Indirizzo MCP	Codice	# byte	Dato 1	Dato 2	...	Dato 256	Check H	Check L
xx	0x14	0x00	0x00	l1	...	0127	xx	xx

Notare che #byte è uguale a 0x00 per indicare 256 byte dati; **ciò vale solo per questo tipo di messaggio** (Codice = 0x14).

Lo stato dei moduli di ingresso sarà contenuto nei byte Dato2÷Dato128, e più precisamente:

Dato2 conterrà lo stato del modulo di ingresso 1,

Dato3 quello del modulo 2, ecc.

Lo stato dei moduli di uscita sarà contenuto nei byte Dato130÷Dato256, e più precisamente

Dato130 conterrà lo stato del modulo di uscita 1,

Dato131 quello del modulo 2 ecc.

Esempio 4: Lettura dello stato dei punti virtuali

Inviare a MCP il messaggio 0x14, specificando nel campo dati, l'indirizzo di partenza della mappa dei punti virtuali e quanti byte si intendono leggere; per leggere tutti i punti virtuali:

Indirizzo di partenza mappa moduli di ingresso: 0x000100 (3 byte)
 Byte da leggere: 128 (1 byte)

Indirizzo MCP	Codice	# byte	Dato 1	Dato 2	Dato 3	Dato 4	Check H	Check L
0x00	0x14	0x04	0x00	0x01	0x00	0x80	0xFF	0x66

MCP risponde con il messaggio contenente 128 byte dati:

Indirizzo MCP	Codice	# byte	Dato 1	Dato 2	...	Dato 128	Check H	Check L
xx	0x14	0x80	0x00	V8-V1	xx	xx

Dato1 non ha significato e quindi va scartato.

Dato2 contiene lo stato dei punti virtuali da 1 a 8 codificato in binario come segue:

bit7	bit6	bit5	bit4	bit3	bit2	bit1	bit0
V8	V7	V6	V5	V4	V3	V2	V1

Dato3 contiene lo stato dei punti virtuali da 9 a 16 codificato in binario come segue:

bit7	bit6	bit5	bit4	bit3	bit2	bit1	bit0
V16	V15	V14	V13	V12	V11	V10	V9

e via via per tutti gli altri byte.

Esempio 5: Lettura della configurazione e dello stato di funzionamento del sistema

La configurazione del sistema è una mappa che indica quali moduli di ingresso e di uscita rientrano nel programma caricato nella memoria di MCP; questi e soltanto questi saranno i moduli che MCP interroga ciclicamente (polling) per verificarne il corretto funzionamento.

MCP mette a disposizione inoltre alcune informazioni riguardanti lo stato di funzionamento "globale" che sono:

- moduli di ingresso guasti e/o scollegati (vale a dire che rientrano nel programma contenuto in MCP ma non rispondono al polling)
- moduli di uscita guasti e/o scollegati (vale a dire che rientrano nel programma contenuto in MCP ma non rispondono al polling)
- moduli di ingresso doppi (vale a dire che nel sistema sono stati collegati due o più moduli che hanno lo stesso indirizzo)
- moduli di uscita doppi (vale a dire che nel sistema sono stati collegati due o più moduli che hanno lo stesso indirizzo)

Per "leggere" la mappa della configurazione e dello stato di funzionamento del sistema, inviare a MCP il messaggio 0x14, specificando nel campo dati, l'indirizzo di partenza della mappa di configurazione del sistema e quanti byte si intendono leggere:

Indirizzo di partenza mappa moduli di ingresso: 0x010140 (3 byte)
 Byte da leggere: 96 (1 byte)

Indirizzo MCP	Codice	# byte	Dato 1	Dato 2	Dato 3	Dato 4	Check H	Check L
0x00	0x14	0x04	0x01	0x01	0x40	0x60	0xFF	0x45

MCP risponde con il messaggio contenente 96 byte dati:

Indirizzo MCP	Codice	# byte	Dato 1	Dato 2	...	Dato 96	Check H	Check L
xx	0x14	0x60	I7÷I1	I15÷I8	xx	xx

Dato1 indica se i moduli di ingresso I1÷I7 sono presenti nella configurazione (bit=1 significa presente, bit=0 significa non presente)

Bit7	bit6	bit5	bit4	bit3	bit2	bit1	bit0
I7	I6	I5	I4	I3	I2	I1	-

Dato2 indica se i moduli di ingresso I8÷I15 sono presenti nella configurazione

Bit7	bit6	bit5	bit4	bit3	bit2	bit1	bit0
I15	I14	I13	I12	I11	I10	I9	I8

e via via fino a Dato16

Dato17 indica se i moduli di uscita O1÷O7 sono presenti nella configurazione (bit=1 significa presente, bit=0 significa non presente)

Bit7	bit6	bit5	bit4	bit3	bit2	bit1	bit0
O7	O6	O5	O4	O3	O2	O1	-

Dato18 indica se i moduli di ingresso O8÷O15 sono presenti nella configurazione

Bit7	bit6	bit5	bit4	bit3	bit2	bit1	bit0
O15	O14	O13	O12	O11	O10	O9	O8

e così via fino a Dato32

In modo analogo:

- da Dato33 a Dato48 sono indicati i moduli di ingresso guasti (bit=1 significa guasto, bit=0 significa non guasto)
- da Dato49 a Dato64 sono indicati i moduli di uscita guasti (bit=1 significa guasto, bit=0 significa non guasto)
- da Dato65 a Dato80 sono indicati i moduli di ingresso doppi (bit=1 significa doppio, bit=0 significa non doppio)
- da Dato81 a Dato96 sono indicati i moduli di uscita doppi (bit=1 significa doppio, bit=0 significa non doppio)

Esempio 6: Comando delle uscite

Inviare a MCP il messaggio 0x16, specificando nel campo dati, l'indirizzo del modulo di uscita da controllare, la maschera che specifica quali punti di uscita di quel modulo vanno controllate, e il nuovo stato delle uscite:

Indirizzo MCP	Codice	# byte	Dato 1	Dato2	Dato3	Check H	Check L
0x00	0x16	0x03	Indirizzo	Maschera	Stato	xx	xx

Indirizzo: indica l'indirizzo del modulo di uscita da 1 a 127

Maschera: indica quali delle 8 uscite devono essere controllate (mediante codifica binaria, vedi esempi che seguono)

Stato: indica il nuovo stato delle uscite (mediante codifica binaria, vedi esempi che seguono)

La maschera permette di agire solo sui punti di uscita specificati (dalla presenza del relativo bit a "1" nella maschera stessa), in modo tale che non ci si debba curare dello stato degli altri punti (che quindi rimangono nello stato in cui si trovavano prima del comando)

Esempi:

se si vuole attivare l'uscita 3 del modulo di indirizzo 6:

Indirizzo MCP	Codice	# byte	Dato 1	Dato2	Dato3	Check H	Check L
0x00	0x16	0x03	0x06	0x04	0x04	0xFF	0xD8

se si vuole disattivare l'uscita 5 del modulo di indirizzo 36:

Indirizzo MCP	Codice	# byte	Dato 1	Dato2	Dato3	Check H	Check L
0x00	0x16	0x03	0x24	0x10	0x00	0xFF	0xB2

se si vuole attivare l'uscita 8 del modulo 2 e contemporaneamente disattivare l'uscita 1:

Indirizzo MCP	Codice	# byte	Dato 1	Dato2	Dato3	Check H	Check L
0x00	0x16	0x03	0x01	0x81	0x80	0xFE	0xE4

Al messaggio di comando delle uscite MCP risponde con il messaggio di acknowledge:

Indirizzo MCP	Codice	# byte	Check H	Check L
xx	0x16	0x00	xx	xx

Il numero di byte 0x00 indica che non ci sono byte dati.

Esempio 7: Comando dei punti virtuali

Inviare a MCP il messaggio 0x19, specificando nel campo dati, il numero del punto virtuale e il corrispondente valore:

Indirizzo MCP	Codice	# byte	Dato 1	Dato2	Check H	Check L
0x00	0x19	0x02	xx	xx

Dato1 e Dato2 formano un numero a 16 bit (Dato1 è byte il più significativo) indicante il numero del punto virtuale (da 1 a 1000). Il bit 15 di tale numero è il valore da assegnare al punto virtuale.

Esempi:

se si vuole attivare il punto virtuale 100:

Indirizzo MCP	Codice	# byte	Dato 1	Dato2	Check H	Check L
0x00	0x19	0x02	0x80	0x64	0xFF	0x00

se si vuole disattivare il punto virtuale 458:

Indirizzo MCP	Codice	# byte	Dato 1	Dato2	Check H	Check L
0x00	0x19	0x02	0x01	0xCA	0xFF	0x19

se si vuole attivare il punto virtuale 458:

Indirizzo MCP	Codice	# byte	Dato 1	Dato2	Check H	Check L
0x00	0x19	0x02	0x81	0xCA	0xFE	0x99

MCP risponde con il messaggio di acknowledge:

Indirizzo MCP	Codice	# byte	Check H	Check L
xx	0x19	0x00	xx	xx

Il numero di byte 0x00 indica che non ci sono byte dati.